

# NAG Library Routine Document

## E01ZMF

### 1 Purpose

E01ZMF generates a multi-dimensional interpolant to a set of scattered data points, using a modified Shepard method. When the number of dimensions is no more than five, there are corresponding routines in Chapter E01 which are specific to the given dimensionality. E01SGF generates the two-dimensional interpolant, while E01TGF, E01TKF and E01TMF generate the three-, four- and five-dimensional interpolants respectively.

### 2 Specification

```
SUBROUTINE E01ZMF (D, M, X, F, NW, NQ, IQ, RQ, IFAIL)
INTEGER          D, M, NW, NQ, IQ(2*M+1), IFAIL
REAL (KIND=nag_wp) X(D,M), F(M), RQ(*)
```

### 3 Description

E01ZMF constructs a smooth function  $Q(\mathbf{x})$ ,  $\mathbf{x} \in \mathbb{R}^d$  which interpolates a set of  $m$  scattered data points  $(\mathbf{x}_r, f_r)$ , for  $r = 1, 2, \dots, m$ , using a modification of Shepard's method. The surface is continuous and has continuous first partial derivatives.

The basic Shepard method, which is a generalization of the two-dimensional method described in Shepard (1968), interpolates the input data with the weighted mean

$$Q(\mathbf{x}) = \frac{\sum_{r=1}^m w_r(\mathbf{x}) q_r}{\sum_{r=1}^m w_r(\mathbf{x})},$$

where  $q_r = f_r$ ,  $w_r(\mathbf{x}) = \frac{1}{\|\mathbf{x} - \mathbf{x}_r\|_2^2}$ .

The basic method is global in that the interpolated value at any point depends on all the data, but E01ZMF uses a modification (see Franke and Nielson (1980) and Renka (1988a)), whereby the method becomes local by adjusting each  $w_r(\mathbf{x})$  to be zero outside a hypersphere with centre  $\mathbf{x}_r$  and some radius  $R_w$ . Also, to improve the performance of the basic method, each  $q_r$  above is replaced by a function  $q_r(\mathbf{x})$ , which is a quadratic fitted by weighted least squares to data local to  $\mathbf{x}_r$  and forced to interpolate  $(\mathbf{x}_r, f_r)$ . In this context, a point  $\mathbf{x}$  is defined to be local to another point if it lies within some distance  $R_q$  of it.

The efficiency of E01ZMF is enhanced by using a cell method for nearest neighbour searching due to Bentley and Friedman (1979) with a cell density of 3.

The radii  $R_w$  and  $R_q$  are chosen to be just large enough to include  $N_w$  and  $N_q$  data points, respectively, for user-supplied constants  $N_w$  and  $N_q$ . Default values of these parameters are provided, and advice on alternatives is given in Section 9.2.

E01ZMF is derived from the new implementation of QSHEP3 described by Renka (1988b). It uses the modification for high-dimensional interpolation described by Berry and Minser (1999).

Values of the interpolant  $Q(\mathbf{x})$  generated by E01ZMF, and its first partial derivatives, can subsequently be evaluated for points in the domain of the data by a call to E01ZNF.

## 4 References

Bentley J L and Friedman J H (1979) Data structures for range searching *ACM Comput. Surv.* **11** 397–409

Berry M W, Minser K S (1999) Algorithm 798: high-dimensional interpolation using the modified Shepard method *ACM Trans. Math. Software* **25** 353–366

Franke R and Nielson G (1980) Smooth interpolation of large sets of scattered data *Internat. J. Num. Methods Engrg.* **15** 1691–1704

Renka R J (1988a) Multivariate interpolation of large sets of scattered data *ACM Trans. Math. Software* **14** 139–148

Renka R J (1988b) Algorithm 661: QSHEP3D: Quadratic Shepard method for trivariate interpolation of scattered data *ACM Trans. Math. Software* **14** 151–152

Shepard D (1968) A two-dimensional interpolation function for irregularly spaced data *Proc. 23rd Nat. Conf. ACM* 517–523 Brandon/Systems Press Inc., Princeton

## 5 Parameters

1: D – INTEGER *Input*

*On entry:*  $d$ , the number of dimensions.

*Constraint:*  $D \geq 2$ .

2: M – INTEGER *Input*

*On entry:*  $m$ , the number of data points.

**Note:** on the basis of experimental results reported in Berry and Minser (1999), when  $D \geq 5$  it is recommended to use  $M \geq 4000$ .

*Constraint:*  $M \geq (D + 1) \times (D + 2)/2 + 2$ .

3: X(D, M) – REAL (KIND=nag\_wp) array *Input*

*On entry:*  $X(1 : D, r)$  must be set to the Cartesian coordinates of the data point  $\mathbf{x}_r$ , for  $r = 1, 2, \dots, m$ .

*Constraint:* these coordinates must be distinct, and must not all lie on the same  $(d - 1)$ -dimensional hypersurface.

4: F(M) – REAL (KIND=nag\_wp) array *Input*

*On entry:*  $F(r)$  must be set to the data value  $f_r$ , for  $r = 1, 2, \dots, m$ .

5: NW – INTEGER *Input*

*On entry:* the number  $N_w$  of data points that determines each radius of influence  $R_w$ , appearing in the definition of each of the weights  $w_r$ , for  $r = 1, 2, \dots, m$  (see Section 3). Note that  $R_w$  is different for each weight. If  $NW \leq 0$  the default value  $NW = \min(2 \times (D + 1) \times (D + 2), M - 1)$  is used instead.

*Suggested value:*  $NW = -1$

*Constraint:*  $NW \leq M - 1$ .

6: NQ – INTEGER *Input*

*On entry:* the number  $N_q$  of data points to be used in the least squares fit for coefficients defining the quadratic functions  $q_r(\mathbf{x})$  (see Section 3). If  $NQ \leq 0$  the default value  $NQ = \min((D + 1) \times (D + 2) \times 6/5, M - 1)$  is used instead.

*Suggested value:*  $NQ = -1$

*Constraint:*  $NQ \leq 0$  or  $(D + 1) \times (D + 2)/2 - 1 \leq NQ \leq M - 1$ .

7: IQ(2 × M + 1) – INTEGER array *Output*

*On exit:* integer data defining the interpolant  $Q(\mathbf{x})$ .

8: RQ(\*) – REAL (KIND=nag\_wp) array *Output*

**Note:** the dimension of the array RQ must be at least  $((D + 1) \times (D + 2)/2) \times M + 2 \times D + 1$ .

*On exit:* real data defining the interpolant  $Q(\mathbf{x})$ .

9: IFAIL – INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

*On entry,*  $D = \langle value \rangle$ .

*Constraint:*  $D \geq 2$ .

*On entry,*  $((D + 1) \times (D + 2)/2) \times M + 2 \times D + 1$  exceeds the largest machine integer.

$D = \langle value \rangle$  and  $M = \langle value \rangle$ .

*On entry,*  $M = \langle value \rangle$  and  $D = \langle value \rangle$ .

*Constraint:*  $M \geq (D + 1) \times (D + 2)/2 + 2$ .

*On entry,*  $NQ = \langle value \rangle$  and  $D = \langle value \rangle$ .

*Constraint:*  $NQ \leq 0$  or  $NQ \geq (D + 1) \times (D + 2)/2 - 1$ .

*On entry,*  $NQ = \langle value \rangle$  and  $M = \langle value \rangle$ .

*Constraint:*  $NQ \leq M - 1$ .

*On entry,*  $NW = \langle value \rangle$  and  $M = \langle value \rangle$ .

*Constraint:*  $NW \leq M - 1$ .

IFAIL = 2

There are duplicate nodes in the dataset.  $X(k, i) = X(k, j)$ , for  $i = \langle value \rangle$ ,  $j = \langle value \rangle$  and  $k = 1, 2, \dots, D$ . The interpolant cannot be derived.

IFAIL = 3

*On entry,* all the data points lie on the same hypersurface. No unique solution exists.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.  
See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.  
See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.  
See Section 3.6 in the Essential Introduction for further information.

## 7 Accuracy

In experiments undertaken by Berry and Minser (1999), the accuracies obtained for a conditional function resulting in sharp functional transitions were of the order of  $10^{-1}$  at best. In other cases in these experiments, the function generated interpolates the input data with maximum absolute error of the order of  $10^{-2}$ .

## 8 Parallelism and Performance

E01ZMF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

E01ZMF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

### 9.1 Timing

The time taken for a call to E01ZMF will depend in general on the distribution of the data points and on the choice of  $N_w$  and  $N_q$  parameters. If the data points are uniformly randomly distributed, then the time taken should be  $O(m)$ . At worst  $O(m^2)$  time will be required.

### 9.2 Choice of $N_w$ and $N_q$

Default values of the parameters  $N_w$  and  $N_q$  may be selected by calling E01ZMF with  $NW \leq 0$  and  $NQ \leq 0$ . These default values may well be satisfactory for many applications.

If non-default values are required they must be supplied to E01ZMF through positive values of NW and NQ. Increasing these parameter values makes the method less local. This may increase the accuracy of the resulting interpolant at the expense of increased computational cost. The default values  $NW = \min(2 \times (D + 1) \times (D + 2), M - 1)$  and  $NQ = \min((D + 1) \times (D + 2) \times 6/5, M - 1)$  have been chosen on the basis of experimental results reported in Renka (1988a) and Berry and Minser (1999). For further advice on the choice of these parameters see Renka (1988a) and Berry and Minser (1999).

## 10 Example

This program reads in a set of 30 data points and calls E01ZMF to construct an interpolating function  $Q(\mathbf{x})$ . It then calls E01ZNF to evaluate the interpolant at a set of points.

Note that this example is not typical of a realistic problem: the number of data points would normally be very much larger.

See also Section 10 in E01ZNF.

## 10.1 Program Text

```

Program e01zmf

!      E01ZMF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: e01zmf, e01znf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Integer                     :: d, i, ifail, liq, lrq, m, n, nq, nw
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: f(:), q(:), qx(:,,:), rq(:), x(:,,:), &
                                xe(:,,:)
Integer, Allocatable        :: iq(:)
!      .. Executable Statements ..
Write (nout,*) 'E01ZMF Example Program Results'

!      Skip heading in data file
Read (nin,*)

!      Input the number of nodes.
Read (nin,*) d, m
liq = 2*m + 1
lrq = (d+1)*(d+2)/2*m + 2*d + 1
Allocate (x(d,m),f(m),iq(liq),rq(lrq))

!      Input the data points X and F.
Do i = 1, m
    Read (nin,*) x(1:d,i), f(i)
End Do

!      Generate the interpolant.
nq = 0
nw = 0

ifail = 0
Call e01zmf(d,m,x,f,nw,nq,iq,rq,ifail)

!      Input the number of evaluation points.
Read (nin,*) n
Allocate (xe(d,n),q(n),qx(d,n))

!      Input the evaluation points.
Do i = 1, n
    Read (nin,*) xe(1:d,i)
End Do

!      Evaluate the interpolant using E01ZNF.
ifail = 0
Call e01znf(d,m,x,f,iq,rq,n,xe,q,qx,ifail)

Write (nout,99998) 'Interpolated Evaluation Points', 'Values'
Write (nout,99997)
Write (nout,99996)(i,i=1,d)
Write (nout,99997)
Write (nout,99999)(i,xe(1:d,i),q(i),i=1,n)

```

```

99999 Format (I4,1X,7F10.4)
99998 Format (/3X,' |',10X,A31,19X,'|',A7)
99997 Format (3X,'---|',60(' - '),'+',7(' - '))
99996 Format (3X,'I |',6(2X,'XE(' ,I1,' ,I)',1X),'|',2X,'Q(I)')
      End Program e01zmf
    
```

### 10.2 Program Data

```

E01ZMF Example Program Data
6 30
      D, M number of dimensions and data points
0.81 0.15 0.44 0.83 0.21 0.64 6.39 X, F data point definition
0.91 0.96 0.00 0.09 0.98 0.37 2.50
0.13 0.88 0.22 0.21 0.73 1.00 9.34
0.91 0.49 0.39 0.79 0.47 0.71 7.52
0.63 0.41 0.72 0.68 0.65 0.83 6.91
0.10 0.13 0.77 0.47 0.22 0.09 4.68
0.28 0.93 0.24 0.90 0.96 0.21 45.40
0.55 0.01 0.04 0.41 0.26 0.79 5.48
0.96 0.19 0.95 0.66 0.99 0.68 2.75
0.96 0.32 0.53 0.96 0.84 0.47 7.43
0.16 0.05 0.16 0.30 0.58 0.90 6.05
0.97 0.14 0.36 0.72 0.78 0.06 0.41
0.96 0.73 0.28 0.75 0.28 0.68 8.68
0.49 0.48 0.58 0.19 0.25 0.67 2.38
0.80 0.34 0.64 0.57 0.08 0.13 3.70
0.14 0.24 0.12 0.06 0.63 0.89 1.34
0.42 0.45 0.03 0.68 0.66 0.17 15.18
0.92 0.19 0.48 0.67 0.28 0.54 4.35
0.79 0.32 0.15 0.13 0.40 0.03 1.50
0.96 0.26 0.93 0.89 0.61 0.81 3.43
0.66 0.83 0.41 0.17 0.09 0.60 3.10
0.04 0.70 0.40 0.54 0.37 0.41 14.33
0.85 0.33 0.15 0.03 0.36 5.77 0.35
0.93 0.58 0.88 0.81 0.40 0.66 4.30
0.68 0.29 0.88 0.60 0.47 0.96 3.77
0.76 0.26 0.09 0.41 0.14 0.30 4.16
0.74 0.26 0.33 0.64 0.36 0.72 6.75
0.39 0.68 0.69 0.37 0.12 0.75 5.22
0.66 0.52 0.17 1.00 0.43 0.19 16.23
0.17 0.08 0.35 0.71 0.17 0.57 10.62 End of data points
6
      N number of evaluation points
0.10 0.10 0.10 0.10 0.10 0.10
      XE evaluation point definition
0.20 0.20 0.20 0.20 0.20 0.20
0.30 0.30 0.30 0.30 0.30 0.30
0.40 0.40 0.40 0.40 0.40 0.40
0.50 0.50 0.50 0.50 0.50 0.50
0.60 0.60 0.60 0.60 0.60 0.60
      End of evaluation points
    
```

### 10.3 Program Results

E01ZMF Example Program Results

I	Interpolated Evaluation Points						Values
	XE(1,I)	XE(2,I)	XE(3,I)	XE(4,I)	XE(5,I)	XE(6,I)	Q(I)
1	0.1000	0.1000	0.1000	0.1000	0.1000	0.1000	-6.5071
2	0.2000	0.2000	0.2000	0.2000	0.2000	0.2000	-3.5708
3	0.3000	0.3000	0.3000	0.3000	0.3000	0.3000	-0.8006
4	0.4000	0.4000	0.4000	0.4000	0.4000	0.4000	1.7257
5	0.5000	0.5000	0.5000	0.5000	0.5000	0.5000	3.9278
6	0.6000	0.6000	0.6000	0.6000	0.6000	0.6000	5.8951